Problem:
Score : 600 points

Problem Statement

We have a tree with N vertices numbered 1 to N.
The i-th edge in this tree connects Vertex a_i and Vertex b_i.

Consider painting each of these edges white or black. There are 2^{N-1} such ways to paint the edges. Among them, how many satisfy all of the following M restrictions?

The i-th (1 \leq i \leq M) restriction is represented by two integers u_i and v_i, which mean that the path connecting Vertex u_i and Vertex v_i must contain at least one edge painted black.

Constraints

2 \leq N \leq 50

1 \leq a_i,b_i \leq N

The graph given in input is a tree.

1 \leq M \leq \min(20,\frac{N(N-1)}{2})

1 \leq u_i < v_i \leq N

If i \not= j, either u_i \not=u_j or v_i\not=v_j

All values in input are integers.

Input

Input is given from Standard Input in the following format:

N
a_1 b_1
:
a_{N-1} b_{N-1}
M
u_1 v_1
:
u_M v_M

Output

Print the number of ways to paint the edges that satisfy all of the M conditions.

Sample Input 1

3
1 2
2 3
1
1 3

Sample Output 1

3

The tree in this input is shown below:

All of the M restrictions will be satisfied if Edge 1 and 2 are respectively painted (white, black), (black, white), or (black, black), so the answer is 3.

Sample Input 2

2
1 2
1
1 2

Sample Output 2

1

The tree in this input is shown below:

All of the M restrictions will be satisfied only if Edge 1 is painted black, so the answer is 1.

Sample Input 3

5
1 2
3 2
3 4
5 3
3
1 3
2 4
2 5

Sample Output 3

9

The tree in this input is shown below:

Sample Input 4

8
1 2
2 3
4 3
2 5
6 3
6 7
8 6
5
2 7
3 5
1 6
2 8
7 8

Sample Output 4

62

The tree in this input is shown below:

Solution in Common Lisp (SBCL 1.1.14) (judge verdict: Accepted)

;; F - Tree and Constraints

(defun main ()
  (let* ((n (read))
         (ab (loop repeat (1- n) collect (cons (read) (read))))
         (m (read))
         (uv (loop repeat m collect (cons (read) (read)))))
    (princ (solve n ab m uv))))

(defun make-adjacency-list (n edges)
  (let ((adj (make-array n :initial-element nil)))
    (loop for i from 0
          for edge in edges
          for a = (1- (car edge))
          for b = (1- (cdr edge))
          do (setf (aref adj a) (cons (cons b i) (aref adj a)))
             (setf (aref adj b) (cons (cons a i) (aref adj b))))
    adj))

(defun find-paths (m uv n adj)
  (labels ((dfs (u v parent)
             (if (eql u v)
                 (empty-set (1- n))
                 (loop for (a . i) in (aref adj u)
                       for p = (and (not (eql a parent))
                                    (dfs a v u))
                       when p do (set-add p i)
                                 (return p)))))
    (loop for (u . v) in uv
          collect (dfs (1- u) (1- v) nil))))

(defun coloring (n paths union sign)
  (if paths
      (+ (coloring n (cdr paths) union sign)
         (coloring n (cdr paths) (set-union union (car paths)) (- sign)))
      (* sign (ash 1 (- (1- n) (count-set-elem union))))))

(defun solve (n ab m uv)
  (let* ((adj (make-adjacency-list n ab))
         (paths (find-paths m uv n adj)))
    (coloring n paths (empty-set (1- n)) 1)))

;; set operations

(defun empty-set (n-elem)
  (make-array n-elem :element-type 'bit :initial-element 0))

(defun set-add (set elem)
  (setf (aref set elem) 1))

(defun set-union (set1 set2)
  (loop with result = (make-array (array-dimension set1 0) :element-type 'bit)
        for b1 across set1
        for b2 across set2
        for i from 0
        do (setf (aref result i) (logior b1 b2))
        finally (return result)))

(defun count-set-elem (set)
  (count 1 set))

(main)